

VOL. 2
NO. 7

STATUS 1500

AUGUST
1984

* 12 issues £10.50 (UK) * overseas £14.50 * published monthly *

* Edited by RONALD COHEN, 62 BLENHEIM CRESCENT, LONDON W.11 *

YOUR SUBSCRIPTIONS

For your subscription, you receive more than merely the newsletter. Your queries - which help keep the newsletter going - receive unstinted help, often requiring much research, or enquiries to subscribers more knowledgeable than myself on some particular subject. Not all of you take advantage of this service, but it is always available. Again, quite a few of you have been put in touch with other users whose interests are similar - not always easy, since subscribers are spread all over the world, and PC 1500 users are not usually gregarious. Is it not therefore obvious that these services are only available to subscribers?

I must take this opportunity to remind some of you that the newsletter is **confidential to the named subscriber**. Shared subscriptions are acceptable, provided that the joint subscribers are named. For groups or corporations, who may find these conditions irksome, anything is negotiable, except silence. I find it sad that the very few subscribers who have been detected cheating on these conditions have invariably been persons in responsible positions.

A further service now available concerns hardware. I receive occasional enquiries about disposing of or acquiring equipment second-hand. So far I have not succeeded in matching buyers and sellers, partly because enquiries are so sporadic. So from now on I shall compile a register of computers and peripherals wanted and unwanted. Let me know what you wish to buy or sell, with minimax prices, and I will try to match any demands. There will be no charge to subscribers for this service.

This issue sees the first of the articles on CE 150 characters in ROM. It will be necessary to read succeeding articles before much use can be made of the technique: unfortunately space prevents the whole system from being exposed in one go.

Both A.E.L.Cox and David Rihoy have very kindly written their articles via SUPERTEXT. They have sent me cassettes, which I can load into the computer, edit and print out immediately. For an editor this is luxury indeed.

STOP PRESS: CE 161 modules (16K) are now available from
MICROS FOR MANAGERS (01-370-5125) at £99.00

59 YOUR SUBSCRIPTIONS
60 SIGNALS
61 PEEK POKE & MEMORY - XVII
62 LETS WRITE A PROGRAM - VII
62 MINDBOGGLE CORNER
63 ROULETTE unfinished

64 ROM CHARACTERS
65 SORT ROUTINES - 1
66 SPREADSHEET 1500 reviewed
67 ELECTRICS - 2
67 VOLTAGE REGULATOR
68 MARKET PLACE

SIGNALS

L.E.SIMONS enquires whether there is any chance of the TRAMSOFT TOOL-2 becoming available in this country.

Most unlikely. **WALTER SPIEDEL** is unwilling to supply in bulk except at more or less retail prices, and therefore no importer can do business with him except at a substantial loss. A number of subscribers have got this module direct from SPIEDEL. If you send him the money he sends the goods immediately, provided he has them in stock. Unfortunately he does not answer enquiries.

R.WILLIAMES believes that it should be possible to use both sides of the CE 158 simultaneously, by accepting data on the PC 1500 via the RS-232 interface in terminal mode, and sending it direct to the printer via the parallel side of the CE 158. He wonders if anyone has actually done this.

I have not had such a feat reported to me, but theoretically it should be possible. Even when the parallel interface is not assigned by OPN "LPRT" data may still be sent via the parallel port to a printer, by the command PRINT #-9,".....etc. However there could be a snag: the RS-232 side may accept data faster than your printer; and when the CE 158 tries to send data to the printer without success for more than 10 seconds ERROR 69 occurs. So you might either have to slow down the intake of data very considerably, or instal a substantial buffer between the CE 158 and the printer.

GERRY LATHAM on a recent visit to Japan found no trace of the PC 1502, or of the GP-40 video interface (see vol.1, page 101). However some interesting other items do exist; they will be imported into USA, though not of course by SHARP (UK). The first is a portable A4 size plotter, CE 515P, for the PC 5000, but usable with PC 1500, at about £150. To use this with the PC 1500 you may need a new interface, the CE 162E, which works with this plotter, and with cassette recorders. A difficulty might arise from the fact that the CE 162E works to JIS (Japan Internal Standard) and not by ASCII codes. The CE 158 can be used instead, for the plotter, but not of course with a cassette-recorder.

I have long suspected that the information from SIMON COX was unreliable. The plotter sounds good value, provided that the cost of the CE 162E is not extortionate. Perhaps when the plotters are current in USA, it may be possible to import a few into UK.

DAVID BOWRING has kindly sent me a translation of parts of SPIEDEL's latest catalogue. There is a MACROASSEMBLER length 4.5K, at DM 178. It is essential to specify what memory module you are using. (It sounds a bit primitive to me). The EXPERIMENTER is a universal interface between PC 1500 and any other peripheral, including even scientific data-gathering equipment. However you would need considerable technical skill to make use of it, as well as DM 648 to pay for it. SPIEDEL also lists a video-interface in his latest catalogue, but there are as yet no details of this.

Thanks for the information. It would be interesting to know, if the interface works with the UK television system.

MIKE SMITH and **GERRY LATHAM** both send details of the PC 1500A. Apparently it only has 8K in the same way that the PC 1500 has 3.75K. The PC 1500A has in fact only 4K more user-RAM than the PC 1500. The memory starts where the standard unexpanded PC 1500 begins, and runs up to 22527.

For an extra 4K it seems hardly worth changing machines, considering the price.

PEEK POKE & MEMORY - XVII

Some light has been shed on one of the queries raised last month, regarding a line consisting of a label alone. It seems that this syntax is supported by ROM version A04, A03, and later versions of A01. It seems that the change took place during the production of A01, without changing its designation. Perhaps this "A01 type 2" is the mysterious A02, about which we know nothing else. The whole affair is a typical piece of SHARP user-unfriendliness. The absence of the revised syntax has caused bugs in the programs of those who take its presence for granted, owing to a familiarity with more conventional versions of BASIC. The revision gives headaches to writers of software using later machines, since the authors do not realise that this syntax will cause ERROR when the software is run on earlier models.

Answered also is my query about the start and end of user-RAM. This information is held in addresses 30819 and 30820, as is shown by the following table:

	CE 161	CE 155	Unexpanded
30819	0	56	64
30820	72	96	72

Multiply the figures by 256 to get the actual addresses. Note that each piece of information can be held in a single byte, because these beginnings and ends are whole numbers of K, and so there is no need for the usual 2-part 256ary.

Thus RESERVE can always be primed simply by:

```
POKE 256*PEEK 30819+86,1,nn,2,nn,.....
```

The beginning of program area is found by $256*PEEK\ 30819+197$

The second piece of information (equivalent to STATUS 3 when there are no 2-character or DIMensioned variables) seems less useful. It ought to be possible, when using CE 161, to POKE 30820,64 so that subsequent DIMensioning of variables causes them to lie entirely within the memory expansion module, thus enabling it to be removed and re-inserted without losing data. But so far I have not found a way of making this work.

ADDRESS-FINDER is now totally obsolete, even without the use of CE 150. If you LIST any line, then the start address of the line is held in 30886 and 30887. From the keyboard:

```
LIST 999
PRINT 256*PEEK 30886+PEEK 30887
```

will give the start of the line, i.e. the address of the high byte of the line number. Unfortunately the command LIST is not supported in program mode. But a little ingenuity still allows a similar technique to be used. The following program line:

```
10: GOTO 999
999: PRINT 256*PEEK 30886+PEEK 30887
```

gives the address of Line 999. In this case however it is not the address of the high byte of the line number that is given, but the address of the first character of the first actual statement of the line. Could not this technique be used in the self-deletion of merged utilities, without (as hitherto) counting back the exact number of bytes to be deleted? Will something like the following work?

```
nnn: GOTO 1
1: POKE 30821,PEEK 30819,197,PEEK 30886,PEEK 30887-3,PEEK 30819,197
```

Slight elaboration may be needed, to take care of occasions when PEEK 30887 is less than 3. Also a flag at the beginning of Line 1, to prevent the utility from being deleted immediately it is RUN.

This article is based mainly on notes supplied by David Rihoy

LETS WRITE A PROGRAM - VII

Overleaf you will find the program as it is at this stage. It can be RUN, though there are still many gaps to fill in. We have not dealt with groups of numbers, or with cancellations. We have not yet written certain restrictions, such as preventing players staking more than their capital, nor dealt with 'breaking the bank'. We still must cope with the graphics, and the tricky questions of cancellations, system betting, etc. There also some decorative touches to add, perhaps: BEEPS, and directions to players. There may be a few bugs; and the correctness of the sums has yet to be checked.

As can be seen, we have not stuck rigidly to our original plan. The pattern is nevertheless recognisable, and there should be no difficulty in finding the right locations for any interpolations. Play is fairly self-explanatory, so no further explanations are given here, except that a new spin of the wheel is induced by responding "ZZZ" to the prompt "name?" for next player. Note that players may stake in any order, and may have more than one bout of staking.

We have not made much use of multi-statement lines, mainly for the sake of clarity; when the program is complete perhaps it can be physically compressed to some extent. However we must avoid putting NEXT or RETURN after INPUT or IF statements, in case they are bypassed. With a little study (and after reading again the previous articles) it should be clear exactly how the program works. If you have any queries please write in as soon as possible, so that I can deal with them in next month's article.

MINDBOGGLE CORNER

'Mindboggle' may take a holiday next month, so here are 3 problems to keep you going. First, a quickie: no prizes. If you key in the following:

4...4...4*1...7...5

what result do you get?

a) 4.44 b) 4 c) 3.33 d) 7.77 e) 0 f) ERROR 1 g) ERROR 9 h) none of these

Try it and see.

The main competition this month is for a program in ONE LINE ONLY. Conditions are stringent. The program when RUN must be self-explanatory. No title, no preface, no footnotes. The purpose of the program may be for use, or for entertainment. But I repeat it must be entirely self-contained. Closing date: October 1st.

As mentioned last month, a prize will also be given for the most adventurous, exotic or unlikely serious use of the PC 1500. Entries should preferably be accompanied by the relevant program, if it is the contestant's own. Closing date: October 15th.

Note that - quite unfairly - these closing dates are provisional, and may be arbitrarily altered according to the number of entries, and the dates of publication of this newsletter.

Of course it is always possible that a single ingenious entry might win both competitions simultaneously.

Last month's competition, for answers to my queries in PEEK & POKE, has been easily won by the notes supplied by DAVID RIHOY.

```

1000 "initialise"DIM A$(7),A(3,7)
3000 "players"INPUT "how many players?";H
3010 FOR F=1TO H
3020 INPUT "name? ";A$(F):INPUT "capital? ";A(1,F):A(1,0)=A(1,0)+A(1,F)
3030 NEXT F
3040 INPUT "alter bank to: ";A(1,0)
3050 CLS :A$(0)="*BANK*"
4000 RESTORE :FOR F=1TO RND 37:READ W:NEXT F
4010 GOSUB 40000
4020 FOR F=0TO H:A(2,F)=0:NEXT F
5000 "stakes"PAUSE " ! PLACE YOUR BETS !"
5200 INPUT " name? ";N$
5205 IF N$="ZZZ"GOTO "spin"
5210 FOR F=1TO H
5220 IF N$=A$(F)GOTO "menu"
5230 NEXT F
5240 BEEP 2,97,97:PAUSE "NAME NOT RECOGNISED-":GOTO 5200
5300 "menu"WAIT 0:PRINT "1/1 2/1 GRP no. nxt CCL"
5305 K=ASC INKEY$:IF K<17OR K>23GOTO 5300
5310 K=K-16:BEEP 1,K*42,300/K:GOSUB 20000+1000*K
5320 GOTO 5300
7000 "spin"PRINT W:COLOR C:CSIZE 7:LF 1:LPRINT " ";W:CSIZE 2:COLOR 2
7200 FOR F=0TO H
7210 A(1,F)=A(1,F)+A(2,F)
7215 Y$="WINS/LOSES"
7220 Z$=LEFT$(Y$,4*(A(2,F)>0))+RIGHT$(Y$,5*(A(2,F)<0))+LEFT$(Y$,10*(A(2,F)=0))
7230 LPRINT A$(F):LPRINT Z$;" ";A(2,F)
7240 LPRINT "capital now=";A(1,F)
7250 LF 1
7260 NEXT F
7270 GOTO 4000
21000 "1/1 menu"WAIT 20:PRINT " red blk odd evn low high":WAIT 0
21005 J=ASC INKEY$:WAIT 0:IF J<17OR J>23GOTO 21000
21010 GOTO 21000+100*(J-16)
21100 RS=1-2*(C<>3):GOSUB 28000:GOSUB 29000:RETURN
21200 RS=1-2*(C=3):GOSUB 28000:GOSUB 29000:RETURN
21300 RS=1-2*(C=1):GOSUB 28000:GOSUB 29000:RETURN
21400 RS=1-2*(E=0):GOSUB 28000:GOSUB 29000:RETURN
21500 RS=1-2*(W>18):GOSUB 28000:GOSUB 29000:RETURN
21600 RS=1-2*(W<19):GOSUB 28000:GOSUB 29000:RETURN
22000 "2/1 menu"PRINT " d1 d2 d3 c1 c2 c3"
22005 J=ASC INKEY$:WAIT 0:IF J<17OR J>23GOTO 22000
22010 GOTO 22000+100*(J-16)
22100 PZ=1:GOSUB 27000:GOSUB 28000:GOSUB 29050:RETURN
22200 PZ=2:GOSUB 27000:GOSUB 28000:GOSUB 29050:RETURN
22300 PZ=3:GOSUB 27000:GOSUB 28000:GOSUB 29050:RETURN
22400 PW=1:GOSUB 27500:GOSUB 28000:GOSUB 29050:RETURN
22500 PW=2:GOSUB 27500:GOSUB 28000:GOSUB 29050:RETURN
22600 PW=3:GOSUB 27500:GOSUB 28000:GOSUB 29050:RETURN
24000 CLS :INPUT "on number..";NU
24010 RS=35*(NU=W)-(NU<>W)
24020 GOSUB 28000:GOSUB 29500:RETURN
25000 BEEP 3:GOTO 5000
27000 RS=2*(PZ=DZ)-(PZ<>DZ):RETURN
27500 RS=2*(CW=PW)-(CW<>PW):RETURN
28000 INPUT "stake? ";S
28010 RETURN
29000 IF W=0LET RS=-0.5
29010 GOTO 29500
29050 IF W=0LET RS=-1
29060 GOTO 29500
29500 A(2,F)=A(2,F)+S*RS:A(2,0)=A(2,0)-S*RS:RETURN
40000 IF W=0LET C=2:C$="0":RETURN
40100 DZ=(W<13)+2*(W>12AND W<25)+3*(W>24)
40200 IF W/3=INT (W/3)LET CW=3
40210 IF (W+1)/3=INT ((W+1)/3)LET CW=2
40220 IF (W+2)/3=INT ((W+2)/3)LET CW=1
40500 "odd/even"E=(W/2=INT (W/2))
41000 "color"D=(W<11OR (W>18AND W<29)):C=3*(D<>E)
41050 C$="BLACK":IF C=3LET C$="RED"
41055 RETURN
60000 "wheel"DATA 0,1,2,3,4,5,6,7,8,9,10,11,12,13,14,15,16,17,18
60010 DATA 19,20,21,22,23,24,25,26,27,28,29,30,31,32,33,34,35,36

```


The purpose of this article is to describe how the CE 150 draws its characters and to show how the machine code routines in ROM can be used to draw other "user defined" characters and some very simple diagrams. For those who do not understand machine code (or who may not wish to) simply plug in the routines provided and use them as directed; but the design of characters/diagrams is up to the user.

Because I happen to think that HEX obscures patterns and is a positive hindrance to understanding, all numbers are given in decimal; in any case I think that you will agree (when you read what follows) that it would be impossible to explain in HEX - OCTAL or BINARY yes, but HEX no.

As will become apparent, there are two major limitations on what can be achieved using this method: (1) lines are drawn in increments set by CSIZE, and (2) lines can only be drawn in one of four directions.

The smallest distance that the pen can move may be called a LINEL, derived from LINE ELEMENT (this is analogous to PIXEL). There are 216 LINELs across the paper and using CSIZE 1 the CE 150 can draw 36 characters; therefore each character is 6 LINELs wide. It is quite easily established that a single character is drawn (apart from any descender) within a square with side=6*CSIZE LINELs. Each character is drawn with a height:width ratio of 6:4. Drawing starts at the bottom LH, and finishes at the bottom RH, corner of the square. The limitation is not that a character has to be within a square, but that it is not possible to draw a line less than CSIZE LINELs.

The other limitation is that it is possible to draw lines only horizontally, vertically, and at 45 degrees to either of these; that is to say, if a square is drawn complete with its diagonals, then these are the only directions possible. That this is so can be easily checked by close examination of all the CE 150 characters. It is important to note that a full diagonal is always drawn: i.e. a line whose length corresponds to the square root of 2 LINELs.

To draw a character requires two things, a pattern and a m/c routine: the next few paragraphs will attempt to explain how the pattern for a character is built up. The table of LPRINT characters in ROM starts at 40960, and the list of pointers to this table starts at 40992. The pattern for each character consists of a varying quantity of numbers (bytes); each number in the pattern contributing one (or more) LINELs to the character: the secret of drawing a character lies in the decoding of each number.

It is the bit pattern of the number which determines what is the direction of a line, how many LINELs it is, and whether the pen is up or down when the line is drawn. The bits decode as follows: the two leftmost bits (MSBs) control pen up/down, the end of drawing, and one other function; the next three bits give the direction; and the last three bits give the length of the line to be drawn. For those to whom bits and HEX are a foreign language, the simplest way of decoding a number is to divide it by 8, the remainder is the length, then divide the answer again by 8; the answer is the control and the remainder is the direction.

It is obvious that the maximum length which can be drawn is 7*CSIZE lineLs. I am going to use the term LINSEG(LINE SEGment) instead of repeating CSIZE LINELs - this will enable me to give concise descriptions of the drawing of characters and diagrams.

For the direction, each remainder from 0 to 7 corresponds to a 45 degree rotation in an anti-clockwise direction. This will be shown in a later table where, for each direction number, will be given the angle in degrees and the co-ordinates (in LINELs) of the equivalent LINE -(x,y) statement.

bit pattern	control	direction	length
	1 1	1 1 1	1 1 1

Table 1

[to be continued

An array of number variables can be sorted into numerical order, and an array of character strings can be sorted into alphabetical order. This is done by ASCII codes, so all numbers heading the character string will come first, then the capital letters A-Z, and after these the lower-case letters a-z.

These notes will deal with character strings only, which present certain complications which do not occur with numerals, but the corresponding numerical program can always be derived from the character string routine.

Sorting is something which the PC 1500, with printer, handles quite efficiently, the main difficulty being to keep running times short. Nearly all the time is expended on repeating the inner loop of the program which must therefore be kept as short as possible. Extra steps in the rest of the program are acceptable if they help to keep the inner loop short, and minimise the number of times it has to be repeated. In the following sorting routines I have assumed that the character strings are stored in a DIMensioned array from A\$(1) to A\$(A), where A is the number of strings ($A < 256$), and that A\$(0) is available as a working register.

Here is the simplest form of BUBBLE sort:

```

100: FOR C=1 TO A-1
110: FOR B=1 TO A-1
120: IF A$(B)>A$(B+1) LET A$(0)=A$(B):
      A$(B)=A$(B+1): A$(B+1)=A$(0)
130: NEXT B: NEXT C: END

```

Table 1

So you can see that between each execution of the outer loop ($C=1,2,\dots$) the inner loop ($B=1,2,\dots$) carries out a pass through all the strings, from bottom ($B=1$ and $B=2$) to the top ($B=A-1$ and $B=A$), testing each pair and changing them over when the lower one exceeds the upper. At the end of the first pass the last string in alphabetical order will have been raised, from whatever position it was in, to its correct position in the top register, A\$(A). At the end of the next pass, the next to last string in alphabetical order will be in the next to top register; and so on until the end of the last pass, when the sorting will be complete.

If there are 101 character strings, the inner loop will be executed 10,000 times. A random-generated set of 101 strings of 16 characters each, which I had prepared, was sorted in 10m20s. Slow! The number of exchanges executed by line 120 was 2895. When the routine was rerun with the items already sorted the time taken was 7m14s; so the 2895 exchanges took 3m6s of the total time, i.e. 30%, and the rate was just over 15 exchanges per second.

However, the routine as it stands contains a serious flaw. When it is used to sort strings exceeding 40 characters in length, it stops immediately and displays ERROR 15 IN 120. This means: "The string buffer size has been exceeded by the character strings handled." In other words the sum of the lengths of the two strings under comparison has exceeded 80. Line 120 must therefore be altered. As only the first 16 characters in either string are subject to comparison by the computer, it can be written as follows whenever the DIM is more than 40:

```

120: IF LEFT$(A$(B),16)>LEFT$(A$(B+1),16) LET ...etc

```

This takes care of strings of all lengths, and I shall include it in the routines which follow as a matter of course; though it increases the running time slightly. Actually when there are 64 (=80-16) characters or less in the strings then only one of the strings being compared need be subjected to ...LEFT\$(A\$(),16)...

The next article in this series will deal with 2 speedier versions of the BUBBLE sort routine.

There are many things I like about this spreadsheet program, and a few which I dislike intensely.

It is beautifully presented, it is simple to use, and the 2nd cassette I received loaded without difficulty. The instructions are clear, and can be learnt in a few minutes. Some useful examples are given, though these do require a certain amount of study. Most operators are supported, not just +/-. An additional option allows so-called 'financial format', so that a million can appear as 1,000,000 instead of 1000000. I wish I were able to take advantage of this facility. The printout is horizontal, and so a whole 13 x 13 cell spreadsheet can be printed out on the CE 150. An excellent device is the use of a key as a toggle switch, so that after recalculating the sheet you may examine any cell either in terms of the formula it contains, or the result which that formula gives. Both absolute and relative replication are supported. A blank spreadsheet is provided for planning, and a pocket card holds reminders of the essential commands and operations.

On the debit side, 2 of the main faults have been cured at the last moment, thus requiring much of this review to be re-written a few hours before publication. The program is now portable, and will work with any memory. The very clever, but slow, EVAL routine has now been discarded in favour of a more primitive but speedier version.

But although the internal operation is now faster, from the user point of view its use is still lamentably slow. A menu must be called, and then responded to, for every single cell entry. With 169 cells, this becomes tedious, although replication facilities can reduce some of the burden. The printout is also unnecessarily slow, the pen moving ponderously across every single one of the 169 cells, even though only a portion of the spreadsheet is actually used.

The author of the program has not previously subscribed to this newsletter, and thus has missed so many of the little tips and tricks which make programming the PC 1500 so very much easier. Against this, he has many most ingenious solutions of his own to problems - but many of these are problems which need not have occurred with a simpler and less ingenious approach. The program is not quite idiot-proof, and the author believes that this cannot be achieved within the limits of the available memory. This may be true of this particular program, but is not so in general. It all depends on how many traps you set! A less clever approach might have set fewer traps, and thus made it easier to protect the user from his own follies. Indeed the obedient user may have no problems: but those whose normal habits are hard to discard may find some difficulties at first: to BREAK with the OFF key, instead of the ON key, may not come automatically to some of us.

You will wish to know how this program compares with EASICALC. If you need to use a spreadsheet constantly, and have the time to learn how to use EASICALC, and do not need operators other than +/-, then for business use you might well prefer that program. If you need the wider range of operators, have a limited budget, need only occasional use, do not have much patience for study, but do on the other hand have plenty of patience for operating, then this SPREADSHEET 1500 may well be what you require.

The author of SPREADSHEET 1500 is now working on a graphics program both for business and scientific use, and we look forward to this with keen interest - though it may be some months before it is completed.

SPREADSHEET 1500 is written by Raymond Scott Grikstas, and is available from PORTABLE COMPUTING, 5 Queens Gardens, Glasgow, at £11.00

When EA 150, CE 150, and PC 1500 are assembled and operated together, then the EA not only charges the batteries in the CE, but joins forces with them to provide the considerable current demanded by the motors for pen and line-feed, which largely determine the interval between recharges. This can be almost indefinitely extended by always using the EA while printing.


If the overall lifetime of the nickel cadmium batteries, as fitted to CE, is to last as long as possible, and premature loss of capacity for recharge is to be avoided, what matters is not the number of times they are discharged and recharged, but how this is done. They respond well to constant current charging at not too high a level. 60 mA is recommended as a maximum for AA cells, and circuitry inside CE takes care of this from a 9v supply. A lower level of current may be beneficial, if you can wait more than 24 hours between recharges. Nicad cells do not mind continuous recharge, but EA should not be left on indefinitely. A charger tailored to provide not more than 10 mA could be used as a permanent trickle charger. A 150 ohm resistor, rated at 0.1 watts or greater, connected in series with one of the low-voltage wires of EA, will convert it from a constant voltage adaptor to a constant current one, providing about 10 mA. It will not now act as a support for printing, but as the trickle-charged batteries will always be full, this only matters if more than 3 hours printing is done in less than 2 days - in which case revert to unmodified EA.

As you know, when CE and PC are connected, and they are turned ON, pen and line-feed activity normally takes place: whereas after AUTO-OFF this activity is avoided when turning ON again. This activity takes more power than the 7mA which flows during the 7 or so minutes of idleness preceding AUTO-OFF. So it would seem that (when CE and PC are connected) allowing AUTO-OFF is more economical than shutting down manually. Tests on AUTO-OFF have shown that it has exactly the same effect as manual OFF, current in both case falling to 30 μ A in the PC, and 130 μ A in the CE.

A final note of caution - a 9v battery connected to CE in place of EA will charge up the internal batteries, and continue to overcharge them to no good purpose until the 9v battery is exhausted. It is wasteful unless the printer is actually in use, AND the internal batteries are low.

VOLTAGE REGULATOR by R.Williames

This circuit diagram runs off a 12-volt NiCad battery pack. It normally delivers about 100mA charging current, but will output up to 1 Amp if required. The battery-pack has 7 Amp-Hours capacity, and could therefore make the user independent of mains supply for several days. The 3-position switch allows current to be supplied at 6, 7.5, or 9 volts.



RESISTORS in OHMS
CAPACITORS 18v WORKING
SWITCH 3 POS SLIDE TYPE
DIODE 1 AMP
ALL PARTS FROM
RADIO SPARES

TWO NEW 'EASI's' from minimicro

Put your thoughts on paper - neatly, dramatically with two new programs from the Minimicro library

en EASI- NOTE 1500

EASI-NOTE is a sophisticated text editor that allows you to use your PC1500/PC2 to rapidly enter text, view it, correct it and print it out in a neatly justified 80-character format in either of 2 type sizes. Features include:-

- Easy text entry with full word-wrap between lines
- View the whole document by automatic scrolling at a speed selected by you - and move to any line in the document easily and quickly
- Insert or delete characters at any point - or rewrite whole lines
- Optional "time-stamping" of text entries for data-logging, laboratory measurements etc
- Storage and recovery of text from tape
- Usable with any RAM module from 4K upwards - and use as much or as little memory as you need
- Ultra-compact coding allows documents of about 300 words to be entered with 4K RAM and over 1000 words with 8K RAM.

ec EASI- CAP 1500

EASI-CAP - previously known as CAP-SET - has been significantly improved to meet Minimicro's exacting standards. It enables you to produce outline capital letters up to the width of the CE-150 printer paper.

- Multiple lines of text can be printed in different sizes and colours, upright or sloping left or right.
- Originally designed by a professional sign-writer to ease his workload, EASI-CAP can, with scissors and paste, be used as a cheap alternative to Letraset to provide dramatic headlines for memos, sales letters, club bulletins - or in any situation where you want your written words to have an impact. Commercial artists are using EASI-CAP to provide rough visuals for their clients.

Both programs feature Minimicro's easy-to-use program structures and detailed documentation. EASI-NOTE requires a minimum of 4K RAM pack. EASI-CAP requires 8K RAM pack as a minimum.

Prices: - EASI-NOTE £19.95 (incl VAT, p&p)
Both programs ordered together £29.95 (incl VAT, p&p)
EASI-CAP £13.95 (incl VAT, p&p)
£29.95 (incl VAT, p&p)

Minimicro software is available from:



Micros for Managers
149 Gloucester Road
LONDON SW7 4TH
Tel 01-370-5125



11 Bury New Road
Prestwich, MANCHESTER M25 8JZ
Tel 061-798-7613

who can also supply

EASI-CALC - The Spreadsheet
EASI-FILE - The Database
EASI-TREND - The Forecaster
EASI-CASH - The Book-keeper
£19.95 each (incl VAT, p&p)

EASI-ONE! - Multi task program
- spreadsheet, mini-texthandler,
statistics...etc, etc

£24.95 (incl VAT, p&p)

If it's SHARP enough to go in your pocket
Minimicro brings it to life!